Facitliste til skriftlig eksamen i Mat1 forårspensum 2021

rev. 180521

Initalisering

```
> restart:with(plots):with(LinearAlgebra):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=(X,Y)->convert(linalg[grad](X,Y),Vector[column]):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)  end 
  proc:
```

MapleTA opgaver (i udvalgt version)

En funktion f af to reelle variable er givet ved

```
> f:=(x,y)->x^2*y-8*x*y+y^2+y+1
```

$$f := (x, y) \mapsto x^2 \cdot y - 8 \cdot x \cdot y + y^2 + y + 1$$

En parametriseret kurve er givet ved

```
> r:=u-><u,1-u^2>
```

$$r := u \mapsto \langle u, 1 - u^2 \rangle$$

Vi betragter den sammensatte funktion h(u) = f(r(u)).

Q1, a

Bestem de to værdier af u for hvilke h'(u) = -8

```
> f(vop(r(u))):
  simplify(%)
```

$$8\, u^3 - 2\, u^2 - 8\, u + 3$$

```
> diff(%,u)
```

$$24\, u^2 - 4\, u - 8$$

```
> solve(%=-8,u)
```

$$0, \frac{1}{6}$$

u1 = 0 og u2 = 1/6.

Q1, b

Angiv koordinaterne til det punkt (x0,y0,z0) som ligger på graffladen for g lodret over punktet r(u2).

```
> r(1/6)
```

$$\begin{bmatrix} \frac{1}{6} \\ \frac{35}{36} \end{bmatrix}$$

```
> f(vop(%))
```

$$\frac{89}{54}$$

x0 = 1/6
y0 = 35/36
z0 = 89/54

En funktion f af to reelle variable er givet ved

```
> f:=(x,y)->x^2*y-4*x^2-8*x*y+y^2+32*x+8*y-44
```

$$f := (x, y) \mapsto x^2 \cdot y - 4 \cdot x^2 - 8 \cdot x \cdot y + y^2 + 32 \cdot x + 8 \cdot y - 44$$

Det oplyses at f har netop ét stationært punkt (x0,y0),

Q2, a

Angiv koordinaterne for det stationære punkt.

```
> lign1:=diff(f(x,y),x)
```

$$lign1 := 2\,x\,y - 8\,x - 8\,y + 32$$

```
> lign2:=diff(f(x,y),y)
```

$$lign2 := x^2 - 8\,x + 2\,y + 8$$

```
> solve({lign1,lign2},{x,y})
```

$$\{x = 4, y = 4\}$$

x0 = 4
y0 = 4

Q2, b

Bestem de partielt afledede af anden orden for f i punktet (x0,y0).

```
> D[1,1](f)(4,4);
  D[1,2](f)(4,4);
  D[2,1](f)(4,4);
  D[2,2](f)(4,4);
```

$$0$$
$$0$$
$$0$$
$$2$$

Q2, c

Vi betragter den begrænsede og afsluttede mængde M som afgrænses af parablen med ligningen y = -x^2+8x-11 og linjen med ligningen y =1.

Bestem den største og mindste værdi som f antager på M.

```
> plot({-x^2+8*x-11,1},x=0..8,view=0..5,scaling=constrained)
```



Hvor skærer de hinanden?

```
> solve(-x^2+8*x-11=1)
```

$$2, 6$$

Det stationære punkt er et indre punkt.

```
> kandidat1=f(4,4)
```

$$kandidat1 = 4$$

Nu følger randundersøgelsen:

Restriktionen af f til parabelstykket:

```
> f(x,-x^2+8*x-11)
```

$$x^2 \left(-x^2 + 8\,x - 11\right) - 12\,x^2 - 8\,x\left(-x^2 + 8\,x - 11\right) + \left(-x^2 + 8\,x - 11\right)^2 + 96\,x - 132$$

```
> simplify(%)
```

$$-x^2 + 8\,x - 11$$

```
> solve(diff(%,x))
```

$$4$$

```
> kandidat2:=f(4,5)
```

$$kandidat2 := 5$$

Restriktionen af f til linjen y=1:

```
> f(x,1)
```

$$-3\,x^2 + 24\,x - 35$$

```
> diff(%,x)
```

$$-6\,x + 24$$

```
> kandidat3:=f(4,1)
```

$$kandidat3 := 13$$

f i skæringspunkterne:

```
> kandidat4=f(2,1);
  kandidat5=f(6,1)
```

$$kandidat4 = 1$$
$$kandidat5 = 1$$

Ved sammenligning af de fem kandidater:
Størsteværdi af f på M er 13 (antages i (4,1))
Mindsteværdi af f på M er 1 (antages i skæringspunkterne (2,1) og (6,1))

En punktmængde A i (x,y)-planen er givet ved

$$A = \{ (x, y) | \ x \in [1, 2] \text{ og } y \in [0, \ln(x)] \}.$$



Der er desuden givet en funktion af to reelle variable ved f(x,y) = x-1

Q4

Bestem planintegralet $\int_A f(x, y)\,d\mu$

p-fremstilling:

```
> r:=(u,v)-><u,0>+v*<0,ln(u)>:
> r(u,v)
```

$$\begin{bmatrix} u \\ v\ln(u) \end{bmatrix}$$

```
> M:=<diff(r(u,v),u)|diff(r(u,v),v)>
```

$$M := \begin{bmatrix} 1 & 0 \\ \frac{v}{u} & \ln(u) \end{bmatrix}$$

```
> Jacobi:=ln(u)
```

$$Jacobi := \ln(u)$$

```
> f:=(x,y)->x-1
```

$$f := (x, y) \mapsto x - 1$$

```
> Int(f(vop(r(u,v)))*Jacobi,[u=1..2,v=0..1])
```

$$\int_0^1 \int_1^2 (u-1)\ln(u)\,du\,dv$$

```
> value(%)
```

$$\frac{1}{4}$$

Planintegralet = 1/4

Q5

I (x,z)-planen i (x,y,z)-rummet betragtes en profilkurve K med ligningen $z = \cosh(x)$, $x \in [0, 1]$.

Lad F betegne den omdregningsflade der fremkommer når K drejes med vinklen pi omkring z-aksen (positiv omløbsretning set fra z-aksens positive ende).

Bestem massen af F med hensyn til massetæthedsfunktionen $f(x, y, z) = \frac{1}{z}, \quad z \neq 0.$

```
> r:=(u,v)-><u*cos(v),u*sin(v),cosh(u)>:
> r(u,v)
```

$$\begin{bmatrix} u\cos(v) \\ u\sin(v) \\ \cosh(u) \end{bmatrix}$$

```
> kryds(diff(r(u,v),u),diff(r(u,v),v)):
  N:=simplify(%)
```

$$N := \begin{bmatrix} -\sinh(u)\,u\cos(v) \\ -\sinh(u)\,u\sin(v) \\ u \end{bmatrix}$$

```
> sqrt(prik(N,N));
  Jacobi:=simplify(%)assuming u::positive
```

$$\sqrt{\sinh(u)^2 u^2 \cos(v)^2 + \sinh(u)^2 u^2 \sin(v)^2 + u^2}$$

$$Jacobi := u\cosh(u)$$

```
> f:=(x,y,z)->1/z:
> Int(f(vop(r(u,v)))*Jacobi,[u=0..1,v=0..Pi])
```

$$\int_0^{\pi} \int_0^1 u\,du\,dv$$

```
> value(%)
```

$$\frac{\pi}{2}$$

Massen = Pi/2

Q7, a

Lad k være et tal større end 0. Der er i (x,y)-planen i (x,y,z)-rummet givet et rektangel B med hjørnerne (0,-1,0), (0,1,0), (k,1,0) og (k,-1,0). Der er endvidere givet højdefunktionen $h(x, y) = k - x \cdot y^2$

På nedenstående figur ses grafen for funktionen h over rektanglet når k = 1.



Bestem for k = 1 rumfanget af det massive legeme Ω som ligger lodret mellem B og grafen for *h* .

```
> k:=1:
> r:=(u,v,w)-><u,v,w*(k-u*v^2)>:
> r(u,v,w)
```

$$\begin{bmatrix} u \\ v \\ w\left(-u\,v^2+1\right) \end{bmatrix}$$

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -w\,v^2 & -2\,w\,u\,v & -u\,v^2+1 \end{bmatrix}$$

```
> Determinant(M):
  Jacobi:=simplify(%)
```

$$Jacobi := -u\,v^2+1$$

```
> Int(Jacobi,[u=0..k,v=-1..1,w=0..1])
```

$$\int_0^1\int_{-1}^1\int_0^1 \left(-u\,v^2+1\right)\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w$$

```
> value(%)
```

$$\frac{5}{3}$$

V = 5/3

Q7, b

Bestem den værdi k skal antages hvis rumfanget af Ω skal være 25/3.

```
> k:='k':
> r:=(u,v,w)-><u,v,w*(k-u*v^2)>:
> r(u,v,w)
```

$$\begin{bmatrix} u \\ v \\ w\left(-u\,v^2+k\right) \end{bmatrix}$$

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -w\,v^2 & -2\,w\,u\,v & -u\,v^2+k \end{bmatrix}$$

```
> Determinant(M):
  Jac:=simplify(%)
```

$$Jac := -u\,v^2+k$$

```
> Int(Jac,[u=0..k,v=-1..1,w=0..1])
```

$$\int_0^1\int_{-1}^1\int_0^k \left(-u\,v^2+k\right)\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w$$

```
> value(%)
```

$$\frac{5\,k^2}{3}$$

```
> solve(%=25/3)[1]
```

$$\sqrt{5}$$

k = sqrt(5)

Essay-opgave (i udvalgt version, jf. pdf)

(det nedenstående er ikke et *essay*, men en udvidet facitliste)

I (x,y,z)-rummet er der givet vektorfelterne

```
> V:=(x,y,z)-><z^2+2*y,3*y-1,x^2+x*z>:
> V(x,y,z)
```

$$\begin{bmatrix} z^2+2\,y \\ 3\,y-1 \\ x^2+x\,z \end{bmatrix}$$

og U(x,y,z) = rot(V)(x,y,z).

```
> U:=(x,y,z)->rot(V)(x,y,z):
> U(x,y,z)
```

$$\begin{bmatrix} 0 \\ z-2\,x \\ -2 \end{bmatrix}$$

```
> P:=<0,1,0>:
  Q:=<-1,0,0>:
  R:=<1,-1,0>:
  S:=<0,0,1>:
```

1)
Linjen fra P til R, hvor u tilhører [0,1]:

```
> s1:=u-><u,-2*u+1,0>:
> s1(u)
```

$$\begin{bmatrix} u \\ -2\,u+1 \\ 0 \end{bmatrix}$$

```
> integrand:=prik(V(vop(s1(u))),diff(s1(u),u))
```

$$integrand := -2+8\,u$$

Det tangentielle kurveintegral:

```
> int(integrand,u=0..1)
```

$$2$$

Linjen fra P til Q, hvor u tilhøreri [0,1]:

```
> s2:=u->P+u*(Q-P):
  s2(u)
```

$$\begin{bmatrix} -u \\ 1-u \\ 0 \end{bmatrix}$$

```
> integrand:=prik(V(vop(s2(u))),diff(s2(u),u))
```

$$integrand := -4+5\,u$$

```
> del1:=int(integrand,u=0..1)
```

$$del1 := -\frac{3}{2}$$

Linjen fra Q til R, hvor u tilhører [0,1]:

```
> s3:=u->Q+u*(R-Q):
  s3(u)
```

$$\begin{bmatrix} -1+2\,u \\ -u \\ 0 \end{bmatrix}$$

```
> integrand:=prik(V(vop(s3(u))),diff(s3(u),u))
```

$$integrand := 1-u$$

```
> del2:=int(integrand,u=0..1)
```

$$del2 := \frac{1}{2}$$

Det tangentielle kurveintegral langs den brudte linje

```
> del1+del2;
```

$$-1$$

V kan ikke være et gradientfelt, da de tangentielle kurveintegraler fra P til Q langs to forskellige veje er forskellige.

2)
Trekanten parametriseret ved rette linjer fra S til punkterne på linjestykket PQ
idet u og v tilhører [0,1]:

```
> r1:=(u,v)->S+v*(s2(u)-S)
```

$$r1 := (u, v) \mapsto S + v \cdot (s2(u) - S)$$

```
> r1(u,v)
```

$$\begin{bmatrix} -u\,v \\ v\,(1-u) \\ 1-v \end{bmatrix}$$

```
> plot3d(r1(u,v),u=0..1,v=0..1,axes=normal,labels=[x,y,z]):
> kryds(diff(r1(u,v),u),diff(r1(u,v),v)):
  N:=simplify(%)
```

$$N := \begin{bmatrix} v \\ -v \\ -v \end{bmatrix}$$

peger nedad!

```
> prik(U(vop(r1(u,v))),N):
  integrand:=expand(%)
```

$$integrand := -2\,u\,v^2 + v^2 + v$$

Fluxen mht til U:

```
> int(integrand,[u=0..1,v=0..1])
```

$$\frac{1}{2}$$

Til sidst bruges Stokes. Hvis man bruger orienteringen PSQP, bliver cirkulationen som det forudgående facit, dvs. 1/2. Med modsat orientering fås -1/2.

3)

Trekanten parametriseres ved rette linjer fra Q til linjen PR

idet u og v tilhører [0,1]:

```
> r2:=(u,v)->Q+v*(s1(u)-Q)
```

$$r2 := (u, v) \mapsto Q + v \cdot (s1(u) - Q)$$

```
> r2(u,v);
```

$$\begin{bmatrix} -1 + v\,(u+1) \\ v\,(-2\,u+1) \\ 0 \end{bmatrix}$$

```
> plot3d(r2(u,v),u=0..1,v=0..1,axes=normal,labels=[x,y,z]):
> kryds(diff(r2(u,v),u),diff(r2(u,v),v)):
  N:=simplify(%)
```

$$N := \begin{bmatrix} 0 \\ 0 \\ 3\,v \end{bmatrix}$$

peger opad som ønsket

```
> prik(V(vop(r2(u,v))),N):
  integrand:=expand(%)
```

$$integrand := 3\,u^2\,v^3 + 6\,u\,v^3 - 6\,u\,v^2 + 3\,v^3 - 6\,v^2 + 3\,v$$

Fluxen mht til V:

```
> int(integrand,[u=0..1,v=0..1])
```

$$\frac{1}{4}$$

4)

Tetraederet parametriseret ved rette linjer fra S til punkterne i trekanten PQR

idet u, v og w tilhører [0,1]:

```
> r3:=(u,v,w)->S+w*(r2(u,v)-S):
> r3(u,v,w)
```

$$\begin{bmatrix} w\,(-1+v\,(u+1)) \\ w\,v\,(-2\,u+1) \\ 1-w \end{bmatrix}$$

```
> M:=<diff~(r3(u,v,w),u)|diff~(r3(u,v,w),v)|diff~(r3(u,v,w),w)>;
```

$$M := \begin{bmatrix} w\,v & w\,(u+1) & -1+v\,(u+1) \\ -2\,w\,v & w\,(-2\,u+1) & v\,(-2\,u+1) \\ 0 & 0 & -1 \end{bmatrix}$$

```
> LinearAlgebra[Determinant](M);
  Jacobi:=-%;
```

$$-3\,w^2\,v$$

$$Jacobi := 3\,w^2\,v$$

```
> divV:=(x,y,z)->div(V)(x,y,z):
> divV(x,y,z)
```

$$3+x$$

```
> integrand:=divV(vop(r3(u,v,w)))*Jacobi;
```

$$integrand := 3\,(3+w\,(-1+v\,(u+1)))\,w^2\,v$$

Ved hjælp af Gauss findes nu fluxen ud gennem overfladen:

```
> int(integrand,[u=0..1,v=0..1,w=0..1])
```

$$\frac{3}{2}$$

5)

```
> lign1:=diff(x(t),t)=0;
  lign2:=diff(y(t),t)=z(t)-2*x(t);
  lign3:=diff(z(t),t)=-2;
```

$$lign1 := \frac{d}{dt}\,x(t) = 0$$

$$lign2 := \frac{d}{dt}\,y(t) = z(t) - 2\,x(t)$$

$$lign3 := \frac{d}{dt}\,z(t) = -2$$

```
> lsn:=dsolve({lign1,lign2,lign3,x(0)=r1(u,v)[1],y(0)=r1(u,v)[2],z
```

```
(0)=r1(u,v)[3]},{x(t),y(t),z(t)});
```

$$lsn := \{x(t) = -u\,v, y(t) = -t^2 + 2\,t\,u\,v + t\,(1-v) - u\,v + v, z(t) = -2\,t + 1 - v\}$$

```
> r:=unapply(<rhs(lsn[1]),rhs(lsn[2]),rhs(lsn[3])>,(u,v,t)):
  'r(u,v,t)'=r(u,v,t);
```

$$r(u, v, t) = \begin{bmatrix} -u\,v \\ -t^2 + 2\,t\,u\,v + t\,(1-v) - u\,v + v \\ -2\,t + 1 - v \end{bmatrix}$$

```
> r(u,v,0),r(u,v,1)
```

$$\begin{bmatrix} -u\,v \\ -u\,v + v \\ 1 - v \end{bmatrix}, \begin{bmatrix} -u\,v \\ u\,v \\ -1 - v \end{bmatrix}$$

```
> plot3d({r(u,v,0),r(u,v,1)},u=0..1,v=0..1,scaling=constrained)
```




```
> N:=kryds(diff(r(u,v,1),u),diff(r(u,v,1),v))
```

$$N := \begin{bmatrix} -v \\ -v \\ 0 \end{bmatrix}$$

```
> sqrt(prik(N,N))assuming v>0:
  Jacobi:=%
```

$$Jacobi := \sqrt{2}\ v$$

```
> Arealet=int(Jacobi,[u=0..1,v=0..1])
```

$$Arealet = \frac{\sqrt{2}}{2}$$